

SIMPLICIAL COMPLEXES IN MACAULAY2

BEN HERSEY, GREGORY G. SMITH, AND ALEXANDRE ZOTINE

ABSTRACT. We highlight some features of the *SimplicialComplexes* package in *Macaulay2*.

This updated version of the *SimplicialComplexes* package in *Macaulay2* [M2], originally developed by Sorin Popescu, Gregory G. Smith, and Mike Stillman, adds constructors for many classic examples, implements a new data type for simplicial maps, and incorporates many improvements to the methods and documentation. Emphasizing combinatorial and algebraic applications, the primary data type encodes an abstract simplicial complex—a family of subsets of a finite set that is closed under taking subsets. These simplicial complexes are the combinatorial counterpart to their geometric realizations formed from points, line segments, filled-in triangles, solid tetrahedra, and their higher-dimensional analogues in some Euclidean space. The subsets in a simplicial complex are called faces. The faces having cardinality 1 are its vertices and the maximal faces (ordered by inclusion) are its facets. The dimension of a simplicial complex is one less than the maximum cardinality of its faces. Following the combinatorial conventions, every nonempty simplicial complex has the empty set as a face.

In this package, a simplicial complex is represented by its Stanley–Reisner ideal. The vertices are identified with a subset of the variables in a polynomial ring and each face is identified with the product of the corresponding variables. A nonface is any subset of the vertices that does not belong to the simplicial complex and each nonface is again identified with a product of variables. The Stanley–Reisner ideal of a simplicial complex is generated by the monomials corresponding to its nonfaces; see Definition 5.1.2 in [BH], Definition 1.6 in [MS], or Definition II.1.1 in [S]. Because computations in the associated polynomial ring are typically prohibitive, this package is not intended for simplicial complexes with a large number of vertices.

CONSTRUCTORS. The basic constructor for a simplicial complex accepts two different kinds of input. Given a list of monomials, it returns the smallest simplicial complex containing the corresponding faces. Given a radical monomial ideal I , it returns the simplicial complex whose Stanley–Reisner ideal is I . We illustrate both methods using the ‘bowtie’ complex in Figure 1.

```
Macaulay2, version 1.20
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               Isomorphism, LLBases, MinimalPrimes, OnlineLookup,
               PrimaryDecomposition, ReesAlgebra, Saturation, TangentCone

i1 : needsPackage "SimplicialComplexes"; S = QQ[v..z];

i3 :  $\Delta$  = simplicialComplex {v*w*x, x*y*z}

o3 = simplicialComplex | xyz vwx |

o3 : SimplicialComplex

i4 : I = monomialIdeal  $\Delta$ 

o4 = monomialIdeal (v*y, w*y, v*z, w*z)

o4 : MonomialIdeal of S
```




FIGURE 1. On the left is the bowtie complex \bowtie and on the right its Alexander dual \bowtie^*

```
i5 :  $\bowtie$  = simplicialComplex I
o5 = simplicialComplex | xyz vwx |
o5 : SimplicialComplex
i6 : assert( $\bowtie$  ==  $\bowtie$ )
```

The package also has convenient constructors for some archetypal simplicial complexes. For example, we recognize the real projective plane and the Klein bottle from the reduced homology groups of some explicit triangulations; see Theorems 6.3–6.4 in [Mu].

```
i7 :  $\mathbb{P}$  = realProjectiveSpaceComplex(2, R = ZZ[a..h])
o7 = simplicialComplex | bef aef cdf adf bcf cde bde ace abd abc |
o7 : SimplicialComplex
i8 : for j from 0 to 2 list prune HH_j  $\mathbb{P}$ 
o8 = {0, cokernel | 2 |, 0}
o8 : List
i9 : for j from 0 to 2 list prune HH_j kleinBottleComplex R
o9 = {0, cokernel | 2 |, 0}
      | 0 |
o9 : List
```

More comprehensively, Frank H. Lutz enumerates simplicial complexes having a small number of vertices; see [L]. Using this list, the package creates a database of 43 138 simplicial 2-manifolds having at most 10 vertices and 1 343 simplicial 3-manifolds having at most 9 vertices. We demonstrate this feature by exhibiting the distribution of f -vectors among the 3-manifolds having 9 vertices. For all nonnegative integers j , the j -th entry in the f -vector is the number of faces having j vertices.

```
i10 : tally for j from 0 to 1296 list fVector smallManifold(3, 9, j, ZZ[vars(1..9)])
o10 = Tally{{1, 9, 26, 34, 17} => 7 }
      {1, 9, 27, 36, 18} => 23
      {1, 9, 28, 38, 19} => 45
      {1, 9, 29, 40, 20} => 84
      {1, 9, 30, 42, 21} => 128
      {1, 9, 31, 44, 22} => 175
      {1, 9, 32, 46, 23} => 223
      {1, 9, 33, 48, 24} => 231
      {1, 9, 34, 50, 25} => 209
      {1, 9, 35, 52, 26} => 121
      {1, 9, 36, 54, 27} => 51
o10 : Tally
```

Exploiting the same loop, we construct the simplicial maps from a minimal triangulation of a torus to the induced subcomplex on the first 7 vertices for each of these 3-manifolds.

```
i11 : T = smallManifold(2, 7, 6, R = ZZ[a..i])
o11 = simplicialComplex | cfg afg beg aeg cdg bdg def bef adf bcf cde ace abd abc |
```

```

o11 : SimplicialComplex
i12 : for j from 0 to 2 list prune HH_j T
o12 = {0, ZZ2, ZZ1}
o12 : List
i13 : for j from 0 to 1296 list (
      phi := map(smallManifold(3, 9, j, R), T, gens R);
      if not isWellDefined phi then continue else phi);
o13 : {}
o13 : List

```

COMBINATORIAL TOPOLOGY. We use the bowtie complex to showcase some of the key operations on simplicial complexes. Viewing a simplicial complex as a subcomplex of a simplex yields a duality theory. For any simplicial complex Δ whose vertices belong to a set V , the Alexander dual is the simplicial complex $\Delta^* := \{F \subseteq V \mid V \setminus F \notin \Delta\}$. Since each simplicial complex in this package has an underlying polynomial ring, the variables in this ring form a canonical superset of the vertices.

```

i14 : dual  $\bowtie$ 
o14 = simplicialComplex | wxz vxz wxy vxy |
o14 : SimplicialComplex
i15 : assert(dual dual  $\bowtie$  ===  $\bowtie$  and dual monomialIdeal  $\bowtie$  === monomialIdeal dual  $\bowtie$ )

```

Algebraically, Alexander duality switches the roles of the minimal generators and the irreducible components in the Stanley–Reisner ideal.

```

i16 : monomialIdeal dual  $\bowtie$ 
o16 = monomialIdeal (v*w, y*z)
o16 : MonomialIdeal of S
i17 : irreducibleDecomposition monomialIdeal  $\bowtie$ 
o17 = {monomialIdeal (v, w), monomialIdeal (y, z)}
o17 : List

```

The topological form of Alexander duality gives an isomorphism between the reduced homology of a simplicial complex and reduced cohomology of its dual; see Theorem 5.6 in [MS].

```

i18 : n = numgens ring  $\bowtie$ 
o18 = 5
i19 : assert all(-1..n-1, j -> prune HH^(n-j-3) dual  $\bowtie$  == prune HH_j  $\bowtie$ )

```

A simplicial complex Δ is Cohen–Macaulay if the associated quotient ring S/I , where I is the Stanley–Reisner ideal of Δ in the polynomial ring S , is Cohen–Macaulay. To characterize this attribute topologically, we introduce a family of subcomplexes. For any face F in Δ , the link is the subcomplex $\text{link}_\Delta(F) := \{G \in \Delta \mid F \cup G \in \Delta \text{ and } F \cap G = \emptyset\}$. The link of the vertex x in \bowtie has two disjoint facets.

```

i20 : L = link( $\bowtie$ , x)

```

```

o20 = simplicialComplex | yz vw |
o20 : SimplicialComplex
i21 : prune HH_0 L
o21 =  $\mathbb{Q}\langle x \rangle$ 
o21 :  $\mathbb{Q}\langle x \rangle$ -module, free

```

As discovered by Gerald Reisner, the simplicial complex Δ is Cohen–Macaulay if and only if, for all faces F in Δ and all integers j less than the dimension of $\text{link}_\Delta(F)$, the j -th reduced homology group of $\text{link}_\Delta(F)$ vanishes; see Corollary 5.3.9 in [BH], Theorem 5.53 in [MS], or Corollary II.4.2 in [S]. Using this criterion, the 0-th reduced homology certifies that \blacktriangleleft is not Cohen–Macaulay.

```

i22 : assert(HH_0 L != 0)
i23 : assert(dim(S^1/monomialIdeal  $\blacktriangleleft$ ) != n - pdim(S^1/monomialIdeal  $\blacktriangleleft$ ))

```

However, the 1-skeleton of \blacktriangleleft is Cohen–Macaulay.

```

i24 :  $\blacktriangleleft$  = skeleton(1,  $\blacktriangleleft$ )
o24 = simplicialComplex | yz xz xy wx vx vw |
o24 : SimplicialComplex
i25 : faceList = rsort flatten values faces  $\blacktriangleleft$ 
o25 = {v*w, v*x, w*x, x*y, x*z, y*z, v, w, x, y, z, 1}
o25 : List
i26 : assert all(faceList, F -> (L := link( $\blacktriangleleft$ , F); all(dim L, j -> HH_j L == 0)))
i27 : assert(dim(S^1/monomialIdeal  $\blacktriangleleft$ ) === n - pdim(S^1/monomialIdeal  $\blacktriangleleft$ ))

```

Alternatively, we verify that \blacktriangleleft is not Cohen–Macaulay by showing that its h -vector has a negative entry; see Theorem 5.1.10 in [BH] or Corollary II.2.5 in [S]. By definition, the h -vector of a simplicial complex Δ is a binomial transform of its f -vector: for all $0 \leq j \leq d := \dim \Delta$, we have $h_j = \sum_{k=0}^j (-1)^{j-k} \binom{d+1-k}{j-k} f_{k-1}$. The h -vector encodes the numerator of the Hilbert series for S/I .

```

i28 : d = dim  $\blacktriangleleft$ 
o28 = 2
i29 : faces  $\blacktriangleleft$ 
o29 = HashTable{-1 => {1}
                0 => {v, w, x, y, z}
                1 => {v*w, v*x, w*x, x*y, x*z, y*z}
                2 => {v*w*x, x*y*z}
o29 : HashTable
i30 : fVec = fVector  $\blacktriangleleft$ 
o30 = {1, 5, 6, 2}
o30 : List
i31 : hVec = for j from 0 to d list
          sum(j+1, k -> (-1)^(j-k) * binomial(d+1-k, j-k) * fVec#k)

```

```

o31 = {1, 2, -1}
o31 : List
i32 : hilbertSeries(S^1/monomialIdeal ◀, Reduce => true)
o32 = 
$$\frac{1 + 2T - T^2}{(1 - T)^3}$$

o32 : Expression of class Divide

```

RESOLUTIONS OF MONOMIAL IDEALS. As David Bayer, Irena Peeva, and Bernd Sturmfels [BPS] reveal, minimal free resolutions of monomial ideals are frequently encoded by a simplicial complex. Consider a monomial ideal J in the polynomial ring $R := \mathbb{Q}[y_1, y_2, \dots, y_m]$. Assume that R is equipped with the \mathbb{N}^m -grading given by $\deg(y_i) = \mathbf{e}_i$, for each $1 \leq i \leq m$, where $\mathbf{e}_1, \mathbf{e}_2, \dots, \mathbf{e}_m$ is the standard basis. Let Δ be a simplicial complex whose vertices are labelled by the generators of J . We label each face F of Δ by the least common multiple $y^{\mathbf{a}_F} \in R$ of its vertices; the empty face is labelled by the monomial $1 = y^{\mathbf{a}_\emptyset}$. The chain complex $C(\Delta)$ supported on the labelled simplicial complex Δ is the chain complex of free \mathbb{N}^m -graded R -modules with basis corresponding to the faces of Δ . More precisely, the chain complex $C(\Delta)$ is determined by the data

$$C_i(\Delta) := \bigoplus_{\dim(F)=i-1} R(-\mathbf{a}_F) \quad \text{and} \quad \partial(F) = \sum_{\dim(G)=\dim(F)-1} \text{sign}(G, F) y^{\mathbf{a}_F - \mathbf{a}_G} G.$$

The symbols F and G represent both faces in Δ and basis vectors in the underlying free module of $C(\Delta)$. The sign of the pair (G, F) belongs to $\{-1, 0, 1\}$ and is part of the data in the boundary map of the chain complex of Δ . For more information, see Subsection 4.1 in [MS] or Chapter 55 in [P].

We illustrate this construction with an explicit example. Consider the simplicial complex Γ in Figure 2 and the monomial ideal $J = (y_0y_1, y_0y_2, y_0y_3, y_1y_2y_3)$ in $R = \mathbb{Q}[y_0, y_1, y_2, y_3]$. Label the



FIGURE 2. On the left is Γ and on the right is the labelling of its vertices

vertices of Γ by the generators of J : $x_0 \mapsto y_0y_1$, $x_1 \mapsto y_0y_2$, $x_2 \mapsto y_0y_3$, and $x_3 \mapsto y_1y_2y_3$.

```

i33 : x = getSymbol "x"; S = ZZ[x_0..x_3];
i35 : Δ = simplicialComplex{x_0*x_1*x_2, x_2*x_3}
o35 = simplicialComplex | x_2x_3 x_0x_1x_2 |
o35 : SimplicialComplex
i36 : chainComplex Δ
o36 = 
$$\begin{array}{cccc} \text{ZZ} & \xleftarrow{1} & \text{ZZ} & \xleftarrow{4} & \text{ZZ} & \xleftarrow{4} & \text{ZZ} & \xleftarrow{1} \\ & -1 & & 0 & & 1 & & 2 \end{array}$$

o36 : ChainComplex
i37 : y = getSymbol "y"; R = QQ[y_0..y_3, DegreeRank => 4];

```

```

i39 : J = ideal(y_0*y_1, y_0*y_2, y_0*y_3, y_1*y_2*y_3)
o39 = ideal (y_0 y_1, y_0 y_2, y_0 y_3, y_1 y_2 y_3)
      0 1    0 2    0 3    1 2 3
o39 : Ideal of R
i40 : C = chainComplex( $\Delta$ , Labels => J_*)
o40 = R  $\xleftarrow{1}$  R  $\xleftarrow{4}$  R  $\xleftarrow{4}$  R  $\xleftarrow{1}$ 
      0      1      2      3
o40 : ChainComplex
i41 : C.dd
o41 = 0 : R  $\xleftarrow{1}$   $\begin{array}{c|cccc} & y_0y_1 & y_0y_2 & y_0y_3 & y_1y_2y_3 \end{array}$  R  $\xleftarrow{4}$  : 1
      1 : R  $\xleftarrow{4}$   $\begin{array}{c|cccc} \{1, 1, 0, 0\} & -y_2 & -y_3 & 0 & 0 \\ \{1, 0, 1, 0\} & y_1 & 0 & -y_3 & 0 \\ \{1, 0, 0, 1\} & 0 & y_1 & y_2 & -y_1y_2 \\ \{0, 1, 1, 1\} & 0 & 0 & 0 & y_0 \end{array}$  R  $\xleftarrow{4}$  : 2
      2 : R  $\xleftarrow{4}$   $\begin{array}{c|ccc} \{1, 1, 1, 0\} & y_3 & & \\ \{1, 1, 0, 1\} & -y_2 & & \\ \{1, 0, 1, 1\} & y_1 & & \\ \{1, 1, 1, 1\} & 0 & & \end{array}$  R  $\xleftarrow{1}$  : 3
o41 : ChainComplexMap
i42 : assert(res(R^1/J) == C)

```

The chain complex $C(\Delta)$ depends on the labelling and is not always a resolution.

```

i43 : C' = chainComplex( $\Delta$ , Labels => reverse J_*)
o43 = R  $\xleftarrow{1}$  R  $\xleftarrow{4}$  R  $\xleftarrow{4}$  R  $\xleftarrow{1}$ 
      0      1      2      3
o43 : ChainComplex
i44 : prune homology C'
o44 = 0 : cokernel | y_0y_3 y_0y_2 y_0y_1 y_1y_2y_3 |
      1 : cokernel {1, 1, 0, 1} | y_2 |
      2 : 0
      3 : 0
o44 : GradedModule

```

Given a monomial ideal J , there are several algorithms that return a labelled simplicial complex Δ such that chain complex $C(\Delta)$ is a free resolution of R/J . We exhibit a few.

```

i45 : J' = monomialIdeal(y_1*y_3, y_2^2, y_0*y_2, y_1^2, y_0^2);
o45 : MonomialIdeal of R
i46 : T = taylorResolution J'

```

```

o46 = R1 <-- R5 <-- R10 <-- R10 <-- R5 <-- R1
      0      1      2      3      4      5

o46 : ChainComplex

i47 : gensJ' = first entries mingens J'

o47 = {y1y32, y22, y0y22, y12, y02}

o47 : List

i48 : S = ZZ[x_0..x_4];

i49 : assert(T == chainComplex(simplexComplex(4, S), Labels => gensJ'))

i50 : assert(lyubeznikSimplicialComplex(J', S) === simplexComplex(4, S))

i51 : Γ = buchbergerSimplicialComplex(J', S)

o52 = simplicialComplex | x_0x_2x_3x_4 x_0x_1x_2x_3 |

o52 : SimplicialComplex

i53 : B = buchbergerResolution J'

o53 = R1 <-- R5 <-- R9 <-- R7 <-- R2
      0      1      2      3      4

o53 : ChainComplex

i54 : assert all(3, i -> HH_(i+1) B == 0)

i55 : assert(betti B == betti res J')

i56 : assert(B == chainComplex(Γ, Labels => first entries mingens J'))

i57 : assert(Γ === lyubeznikSimplicialComplex(J', S, MonomialOrder => {2,1,0,3,4}))

i59 : assert(Γ === scarfSimplicialComplex(J', S))

```

For more information about the Taylor resolution, the Lyubeznik resolution, and the Scarf complex, see [Me]. The Buchberger resolution is described in [OW].

ACKNOWLEDGEMENTS. All three authors were partially supported by the Natural Sciences and Engineering Research Council of Canada (NSERC).

REFERENCES

- [BH] Winfried Bruns and Jürgen Herzog, *Cohen-Macaulay Rings*, Cambridge Studies in Advanced Mathematics, vol. 39, Cambridge University Press, 1993.
- [BPS] Dave Bayer, Irena Peeva, and Bernd Sturmfels, *Monomial resolutions*, Math. Res. Lett. **5** (1998), no. 1-2, 31–46.
- [L] Frank H. Lutz, *The Manifold Page* (2017), <http://page.math.tu-berlin.de/~lutz/stellar/>.
- [M2] Daniel R. Grayson and Michael E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at <http://www.math.uiuc.edu/Macaulay2/>.
- [Me] Jeff Mermin, *Three simplicial resolutions*, Progress in Commutative Algebra 1, De Gruyter, 2012, pp. 127–141.
- [MS] Ezra Miller and Bernd Sturmfels, *Combinatorial Commutative Algebra*, Graduate Texts in Mathematics, vol. 227, Springer-Verlag, 2005.
- [Mu] James R. Munkres, *Elements of Algebraic Topology*, Addison-Wesley Publishing, 1984.

- [OW] Anda Olteanu and Volkmar Welker, *The Buchberger resolution*, J. Commut. Algebra **8** (2016), no. 4, 571–587.
- [P] Irena Peeva, *Graded Syzygies*, Algebra and Applications, vol. 14, Springer-Verlag, 2011.
- [S] Richard P. Stanley, *Combinatorics and commutative algebra*, 2nd ed., Progress in Mathematics, vol. 41, Birkhäuser, 1996.

DEPARTMENT OF MATHEMATICS & STATISTICS, CONCORDIA UNIVERSITY, MONTRÉAL, QB, H3G 1M8, CANADA
benjamin.hersey@concordia.ca.

DEPARTMENT OF MATHEMATICS & STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ON, K7L 3N6, CANADA
ggsmith@mast.queensu.ca, 18az45@queensu.ca.